

INTRODUCCIÓN A WIKIDATA

Jose Emilio Labra Gayo

WESO - Web Semantics Oviedo

Wikidata (<http://wikidata.org/>)

- Forma parte de Wikimedia
- Wikimedia = organización que da soporte a Wikipedia
- Wikidata: Comienza en 2012
- Conocimiento estructurado
- Editable por humanos/bots
- 94 millones de datos (Marzo, 2021)




Objetivos



- Aglutinar el conocimiento humano como datos enlazados
 - Dar soporte a Wikipedia
 - Soporte a enlaces interwiki
 - Soporte a cajas de información
 - Conocimiento estructurado
 - Consultas enriquecidas
 - ...y a muchos otros proyectos
 - Licencia libre: CC0

Enunciados

- Modelo de datos basado en enunciados
- Enunciado = Tripleta (Sujeto - Predicado - Objeto)



Entidades y propiedades

Se representan mediante URIs = identificador global

Permite obtener más información sobre ellos al de-referenciarlos

Esquema de nombres opacos (se identifican con números)

Independiente de cualquier idioma

Entidades empiezan por Q

Ejemplo:

Oviedo: <http://www.wikidata.org/entity/Q14317>

Asturias: <http://www.wikidata.org/entity/Q3934>

Propiedades empiezan por P

Ejemplo:

"Es capital de": <https://www.wikidata.org/prop/direct/P1376>

"lugar de nacimiento": <https://www.wikidata.org/prop/direct/P19>

Los enunciados relacionan URIs



Se representa como



Prefijos

Los prefijos permiten simplificar las URIs largas



Si asociamos

wd = <http://www.wikidata.org/entity/>

wdt = <https://www.wikidata.org/prop/direct/>


Entonces puede escribirse como:



NOTA:


Prefijos comunes en Wikidata: <https://wikibooks.org/wiki/SPARQL/Prefijos>

Grafo de conocimiento




Los objetos pueden ser literales

- Los objetos también pueden ser números, texto, fechas, imágenes, etc...




Referencias o fuentes

- Permite justificar el origen del enunciado
 - ▣ Se considera buena práctica incluir referencias siempre
- Propiedades habituales
 - ▣ wdt:P248 (stated in): para publicaciones
 - ▣ wdt:P854 (reference URL): para sitios web




Cualificadores

- Permiten anotar o contextualizar enunciados
 - ▣ Ejemplo, la población de Oviedo depende del año



Rangos

- Cuando un enunciado tiene multiples valores, los rangos (ranks) permiten indicar:
 - ▣ Cuál es el valor preferido
 - ▣ Cuál es el valor normal
 - ▣ Cuál es el valor obsoleto (*deprecated*)






Datos obsoletos o erróneos

- Se recomienda no borrar enunciados
 - ▣ Los enunciados con valores obsoletos o erróneos pueden marcarse como obsoletos (deprecated)
 - ▣ Varias razones para marcar enunciados obsoletos
 - No puede confirmarse en las fuentes
 - Identificador retirado
 - Valor incorrecto
 - Fecha de fallecimiento en persona viva
 - ...

Etiquetas, descripciones y alias

- Las etiquetas (labels) proporcionan un texto breve para identificar la entidad o propiedad
 - ▣ El texto puede estar en múltiples idiomas
- Las descripciones pueden extender las etiquetas
- Los alias permiten dar nombres alternativos



Entidades ó items

- Permiten representar cualquier cosa o concepto
- Tienen identificadores únicos
 - ▣ Deben representar algo claramente identificable

Propiedades

- Las propiedades en Wikidata permiten integrar conocimiento
 - ▣ Lista con propiedades ya existentes
 - https://www.wikidata.org/wiki/Wikidata:List_of_properties
 - ▣ Para añadir propiedad nueva, realizar propuesta
 - Debe ser aprobada (y discutida) por la comunidad

Propiedades para relacionar ítems

- Es una instancia de ... (P31)
- Es el mismo que ... (P460)
- Es diferente, pero a veces se confunde con ... (P1881)
- Es parte de ... (P361)
- Contiene ... (P527)
- ...

Edición en Wikidata

- La edición de contenidos puede ser manual o mediante bots
- Edición manual es similar a Wikipedia
 - ▣ Click en Edit y a jugar!!!
 - ▣ Buenas prácticas:
 - Identificarse mediante "login"
 - Añadir referencias
 - Añadir cualificadores
- Antes de editar se recomienda practicar en:
 - ▣ <https://test.wikidata.org/>
 - ▣ <https://www.wikidata.org/wiki/Wikidata:Sandbox>

Consultas en Wikidata


- Wikidata Query Service
 - ▣ Servicio de consultas de Wikidata
 - ▣ Lenguaje de consultas: SPARQL
 - ▣ <https://query.wikidata.org/>

Consulta básica

- "Personas que nacen en Oviedo"
 - ▣ Los valores que empiezan por ? Son variables

```
SELECT ?item WHERE {  
  ?item wdt:P19 wd:Q14317 .  
}
```

<http://tinyurl.com/y35wndba>



Etiquetas

- Wikidata tiene un servicio para generar etiquetas
 - ▣ **SERVICE** wikibase:label { ... }
 - ▣ Se genera etiqueta para los items ?XXX que se incluyan en SELECT como ?xxxLabel

```
SELECT ?item ?itemLabel {  
  
    ?item wdt:P19 wd:Q14317 .  
  
    Generador de etiquetas  
  
    SERVICE wikibase:label {  
        bd:serviceParam wikibase:language "[AUTO_LANGUAGE],es".  
    }  
}
```

[AUTO_LANGUAGE] = Idioma configurado en navegador

Buscando más valores

- Personas que nacen en Oviedo y fecha de nacimiento

```
SELECT ?item ?itemLabel ?fechaNacimiento {  
  
  ?item wdt:P19 wd:Q14317 .  
  ?item wdt:P569 ?fechaNacimiento  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],es".  
  }  
}
```

<http://tinyurl.com/y24umlhy>

IMPORTANTE: Si se ponen varios patrones de enunciados, separarlos con punto (.)

Ordenando resultados

- ORDER BY permite ordenar resultados
 - *"Personas que nacen en Oviedo ordenadas por fecha de nacimiento"*

```
SELECT ?item ?itemLabel ?fechaNacimiento {  
  
  ?item wdt:P19 wd:Q14317 .  
  ?item wdt:P569 ?fechaNacimiento  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],es".  
  }  
} ORDER BY ?fechaNacimiento
```

Orden descendente resultados

ORDER BY DESC ordena resultados descendientemente

"Personas que nacen en Oviedo ordenadas por fecha de nacimiento (primero el más joven)"

```
SELECT ?item ?itemLabel ?fechaNacimiento {  
  
  ?item wdt:P19 wd:Q14317 .  
  ?item wdt:P569 ?fechaNacimiento  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],es".  
  }  
} ORDER BY DESC(?fechaNacimiento)
```

Patrones de búsqueda

- Las variables pueden ponerse en cualquier sitio de la tripleta
 - ▣ Ejemplo: Relación entre Oviedo y Asturias

```
SELECT ?relacion {  
  wd:Q14317 ?relacion wd:Q3934 .  
}
```

- ▣ Capitales de regiones en España

<http://tinyurl.com/y29gfhdg>

```
SELECT ?ciudad ?ciudadLabel ?region ?regionLabel {  
  ?ciudad wdt:P1376 ?region .  
  ?region wdt:P17 wd:Q29  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],es".  
  }  
}
```

Simplificando patrones con ;

- Punto y coma (;) permite omitir sujetos repetidos
- Futbolistas nacidos en Oviedo

```
SELECT ?person ?personLabel {  
  ?person wdt:P19 wd:Q14317 ;  
          wdt:P106 wd:Q937857 .  
  
SERVICE wikibase:label {  
  bd:serviceParam wikibase:language "[AUTO_LANGUAGE],es".  
}  
}
```

```
?person wdt:P19 wd:Q14317 ;  
        wdt:P106 wd:Q937857 .
```

=

```
?person wdt:P19 wd:Q14317 .  
?person wdt:P106 wd:Q937857 .
```

Simplificando patrones con ,

- Cuando se repite el sujeto y predicado se pueden omitir mediante ,
- Futbolistas y actores de nacionalidad española

```
SELECT ?person ?personLabel {  
  ?person wdt:P27 wd:Q29 .  
  
  ?person wdt:P106 wd:Q937857, wd:Q33999 .  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],es".  
  }  
}
```

```
?person wdt:P106 wd:Q937857, wd:Q33999 .
```

=

```
?person wdt:P106 wd:Q937857 .  
?person wdt:P106 wd:Q33999 .
```

Filtros

- Condiciones que devuelven true/falso
- Se incluyen los resultados si se cumple la condición
 - ▣ Ejemplo capitales de regiones con más de 20000 habitantes

```
SELECT ?ciudad ?ciudadLabel ?region ?regionLabel {
  ?ciudad wdt:P1376 ?region .
  ?region wdt:P17 wd:Q29 .
  ?region wdt:P1082 ?poblacion
  FILTER (?poblacion > 20000)
  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],es".
  }
}
```

Tipos de valores en comparaciones

- Booleanos
 - ▣ Valores: `true`, `false`
 - ▣ Operadores: `&&` (and), `||` (or)
- Números
 - ▣ Operadores: `=`, `<`, `>`, `>=`, `<=`, `!=`
- Cadenas de texto
 - ▣ Se representan entre comillas dobles `"Asturias"`
 - ▣ Con idioma: `"Universidad de Oviedo"@es`
- Fechas
 - ▣ Ejemplo: `"2019-03-01"^^xsd:dateTime`

Comparaciones con fechas

- Personas que nacen en Oviedo entre 1965 y 1970
 - ▣ Las fechas se ponen como "1965-01-01"^^xsd:dateTime

```
SELECT ?item ?itemLabel ?fechaNacimiento {  
  
  ?item wdt:P19 wd:Q14317 .  
  ?item wdt:P569 ?fechaNacimiento  
  
  FILTER("1965-01-01"^^xsd:dateTime <= ?fechaNacimiento &&  
         ?fechaNacimiento < "1970-01-01"^^xsd:dateTime).  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],es".  
  }  
} ORDER BY DESC(?fechaNacimiento)
```

Mostrar resultados en imágenes

- Servicio ImageGrid
 - ▣ Un comentario puede sugerir que se vea por defecto
- Fotos de personas nacidas en Oviedo junto con fecha de nacimiento

```
#defaultView: ImageGrid

SELECT ?person ?personLabel ?imagen ?fechaNacimiento {
  ?person wdt:P19 wd:Q14317 .
  ?person wdt:P18 ?imagen .
  ?person wdt:P569 ?fechaNacimiento
  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".
  }
} ORDER BY DESC(?fechaNacimiento)
```

Líneas temporales

- Variables temporales
- Personas nacidas en Oviedo a lo largo del tiempo

```
#defaultView:Timeline
```

```
SELECT ?person ?personLabel ?imagen ?fechaNacimiento {  
  ?person wdt:P19 wd:Q14317 .  
  ?person wdt:P18 ?imagen .  
  ?person wdt:P569 ?fechaNacimiento  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".  
  }  
} ORDER BY DESC(?fechaNacimiento)
```

Vista de mapa

- Personas nacidas en Oviedo junto con su lugar de fallecimiento en un mapa

```
#defaultView:Map
SELECT ?person ?personLabel ?imagen
      ?lugarFallecimiento ?lugarFallecimientoLabel
      ?coordenadasFallecimiento {
  ?person wdt:P19 wd:Q14317 .
  ?person wdt:P18 ?imagen .
  ?person wdt:P569 ?fechaNacimiento .
  ?person wdt:P20 ?lugarFallecimiento .
  ?lugarFallecimiento wdt:P625 ?coordenadasFallecimiento
  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".
  }
} order by ?fechaNacimiento
```

Etiquetas de forma tradicional

□ Usando servicio wikibase:label

```
select ?person ?personLabel where {  
  ?person wdt:P19 wd:Q14317 .  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "es".  
  }  
}
```

<http://tinyurl.com/yxlczbek>

□ Método tradicional

```
select ?person ?personLabel where {  
  ?person wdt:P19 wd:Q14317 ;  
  rdfs:label ?personLabel  
  FILTER (Lang(?personLabel) = 'es')  
}
```

<http://tinyurl.com/yxrwgq9j>

Lang(...) = devuelve el idioma de una cadena de texto

Filtros por etiqueta

- Para realizar filtros por etiquetas debe usarse el método tradicional
 - ▣ Personas nacidas en Oviedo cuyo nombre empieza por "José"

```
SELECT ?personLabel WHERE {  
  ?person wdt:P19 wd:Q14317 .  
  ?person rdfs:label ?personLabel .  
  FILTER (Lang(?personLabel)="es") .  
  FILTER (strstarts(?personLabel,"José"))  
}
```

NOTA: El servicio de generación de etiquetas se invoca al terminar la consulta y sus valores no están accesibles

Funciones con cadenas de texto

□ Librería de funciones de Xpath

`strlen(str)` = longitud de str

`ucase(str)` convierte a mayúsculas

`lcase(str)` convierte a minúsculas

`substr(str,inicio,tam?)` = subcadena a partir de inicio de tamaño tam

`substr('camino',3,2)='mi'`

`strstarts(str1,str2)` = true si str1 comienza con str2

`strends(str1,str2)` = true si str1 finaliza con str2

`contains(str1,str2)` = true si str1 contiene str2

`encode_for_uri (str)` = resultado de codificar str

`concat (str1,...strN)` = concatenación de cadenas

`langMatches(str,lang)` = true si encaja el idioma

`regex(str,patrón,flags)` = true si encaja la expresión regular

Regex: Expresiones regulares

Utiliza la función de XPath 2.0

`regex(?Expresión, ?Patrón [, ?Flags])`

?Expresión = expresión a encajar

?Patrón = expresión regular con la que se encaja

?Flags = opciones para el encaje

`^` = Inicio de cadena
`$` = Fin de la cadena
`.` = Cualquier carácter
`\d` = dígito
`?` = opcional, `*` = 0 ó más, `+` = 1 ó más
`X{n}` = encaja X n veces
`X{m,n}` = encaja X de m a n veces

Flags:

`i` = insensible mayúsculas/minúsculas
`m` = múltiples líneas
`s` = línea simple
`x` = elimina espacios en blanco

Regex: expresiones regulares

- Ejemplo: personas que necen en Oviedo y cuyo number contiene la expresión "bra"

```
SELECT ?personLabel WHERE {  
  ?person wdt:P19 wd:Q14317 .  
  ?person rdfs:label ?personLabel .  
  FILTER (Lang(?personLabel)="es") .  
  FILTER (regex(?personLabel,"bra","i"))  
}
```

Funciones numéricas

`abs(n)` = valor absoluto

`floor(n)` = redondear n° hacia bajo

`round(n)` = redondear n°

`ceil(n)` = redondear n° hacia arriba

`rand()` = n° aleatorio entre 0 y 1

Funciones con fechas

`now()` = devuelve el instante actual

`year(i)` = devuelve el año de un instante de tiempo `i`

`year("2011-01-10T14:45:13.815-05:00"^^xsd:dateTime) = 2011`

`month(i)` = devuelve el mes de `i`

`month("2011-01-10T14:45:13.815-05:00"^^xsd:dateTime) = 1`

`day(i)` = devuelve el día de `i`

`day("2011-01-10T14:45:13.815-05:00"^^xsd:dateTime) = 10`

`hours(i)` = devuelve la hora de `i`

`hours("2011-01-10T14:45:13.815-05:00"^^xsd:dateTime) = 14`

□ `minutes(i)` = devuelve los minutos de `i`

▣ `minutes("2011-01-10T14:45:13.815-05:00"^^xsd:dateTime) = 45`

□ `seconds(i)` = devuelve los segundos de `i`

▣ `seconds("2011-01-10T14:45:13.815-05:00"^^xsd:dateTime) = 13.815`

□ `timezone(i)` = devuelve la zona temporal de `i`

▣ `timezone("2011-01-10T14:45:13.815-05:00"^^xsd:dateTime) = -PT5H`

□ `tz(i)` = devuelve la zona temporal de `i`

▣ `tz("2011-01-10T14:45:13.815-05:00"^^xsd:dateTime) = -5`

Funciones HASH

`md5(str)` = aplica el algoritmo MD5 a `str`

`sha1(str)`, `sha224(str)`, `sha256(str)`,

`sha384(str)`, `sha512(str)` = calculan el valor HASH de `str` utilizando las variaciones correspondientes del algoritmo SHA

Unión de grafos

- UNION combina resultados de varios grafos

Ejemplo: Personas nacidas en Oviedo que son pintores o actores

```
SELECT ?persona ?personaLabel {
  ?persona wdt:P19 wd:Q14317 .
  { ?persona wdt:P106 wd:Q1028181 . }
  UNION
  { ?persona wdt:P106 wd:Q33999 }

  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".
  }
}
```

Encajes opcionales

- OPTIONAL permite obtener valores sin fallar cuando algunos valores no existen

Pintores que nacen en Oviedo con su fecha de nacimiento y de fallecimiento

```
SELECT ?persona ?personaLabel ?fechaNacimiento ?fechaFallecimiento {
  ?persona wdt:P19 wd:Q14317 ;
           wdt:P106 wd:Q1028181 ;
           wdt:P569 ?fechaNacimiento .
  OPTIONAL { ?persona wdt:P570 ?fechaFallecimiento }

SERVICE wikibase:label {
  bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".
}
}
```

Control de resultados

DISTINCT elimina valores duplicados

ORDER BY permite especificar el orden de los resultados (puede especificarse ASC, DESC...)

LIMIT n indica el número de resultados

OFFSET m indica a partir de qué resultado empezar a contar

Expresiones PATH

- Permiten ajustar valores de las propiedades
- Posibilidades:
 - / = secuencia
 - | = alternative
 - * = repetición 0 ó más
 - + = repetición 1 ó más
 - ? = opcional o ó 1

Expresiones PATH: secuencia /

- / representa secuencia
- Ejemplo: hijo de hijo = wdt:P40/wdt:P40
- Nietos de pintores que también son pintores

```
SELECT ?persona ?personaLabel ?nieto ?nietoLabel {  
  
  ?persona wdt:P40/wdt:P40 ?nieto .  
  
  ?persona wdt:P106 wd:Q1028181 .  
  ?nieto wdt:P106 wd:Q1028181 .  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".  
  }  
}
```

Expresiones PATH: repetición *

- * representa 0 o más
 - ▣ Ejemplo: wdt:P40* (descendientes incluido el nodo)
 - ▣ Descendientes de Picasso que son pintores
 - Incluido Picasso

```
SELECT ?persona ?personaLabel ?desc ?descLabel {  
  
  wd:Q5593 wdt:P40* ?desc .  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".  
  }  
}
```

Expresiones PATH: repetición +

- + representa 1 o más
 - ▣ Ejemplo: wdt:P40+ (descendientes sin incluir el nodo)
 - ▣ Descendientes de Picasso que son pintores
 - Sin incluir a Picasso

```
SELECT ?persona ?personaLabel ?desc ?descLabel {  
  
  wd:Q5593 wdt:P40+ ?desc .  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".  
  }  
}
```

Expresiones PATH: Alternativa |

- | expresa alternativa
 - ▣ Ejemplo: progenitor de Picasso

```
SELECT ?persona ?personaLabel ?progenitor ?progenitorLabel {  
  wd:Q5593 wdt:P22|wdt:P25 ?progenitor .  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".  
  }  
  
}
```

Expresiones PATH combinadas

□ Ejemplo: ascendentes de Felipe VI

```
SELECT ?ascendente ?ascendenteLabel {  
  
  wd:Q191045 (wdt:P22|wdt:P25)* ?ascendente .  
  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".  
  }  
  
}
```

Instancias y subclases

- Propiedad P31 (instancia de)
- Propiedad P279 (subclase)
- Todos los elementos de un conjunto: P31/P279*
 - ▣ Instancias de una clase o de una subclase de una clase...
- Todos los movimientos artísticos

```
SELECT ?movimiento ?movimientoLabel {
  ?movimiento wdt:P31/wdt:P279* wd:Q968159

  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".
  }
}
```

BIND

- BIND Permite asignar un valor a una variable
- Ejemplo calcular edad de personas de Oviedo que fallecen en Oviedo

```
SELECT ?persona ?personaLabel ?edad {  
  ?persona wdt:P19 wd:Q14317 .  
  ?persona wdt:P569 ?fechaNacimiento .  
  ?persona wdt:P570 ?fechaFallecimiento .  
  BIND(FLOOR((?fechaFallecimiento - ?fechaNacimiento)/365) AS ?edad).  
  SERVICE wikibase:label {  
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".  
  }  
}
```

Agrupaciones



- GROUP BY permite agrupar valores
- Funciones de agregación: COUNT, MAX, MIN, SUM,...
- Ejemplo: Listado de países con número de pintores

```
SELECT ?país ?paísLabel (COUNT (DISTINCT ?pintor) AS ?númeroPintores)
WHERE {
  ?pintor wdt:P106 wd:Q1028181 .
  ?pintor wdt:P27 ?país .

  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".
  }
}
GROUP BY ?país ?paísLabel
ORDER BY DESC (?númeroPintores)
```

Enunciados cualificados

- Cada propiedad de un enunciado cualificado tiene un valor ficticio que puede contener otros enunciados



Poblaciones de Oviedo

```
SELECT ?población ?año WHERE {  
  wd:Q14317 p:P1082 ?cualificado .  
  ?cualificado ps:P1082 ?población .  
  ?cualificado pq:P585 ?año .  
} ORDER BY ?año
```

Nodos anónimos

- Nodos anónimos son nodos auxiliares cuyos valores no nos interesan
- Pueden escribirse como `_:etiqueta`
- `⊙` encerrados entre corchetes `[...]`

Población de Oviedo por años

```
SELECT ?población ?año WHERE {  
  wd:Q14317 p:P1082 [  
    ps:P1082 ?población ;  
    pq:P585 ?año  
  ].  
} ORDER BY ?año
```

Información errónea

- El valor del rango wikibase:DeprecatedRank indica que un enunciado se considera incorrecto

Fechas de fallecimiento incorrectas de personas españolas

```
select ?person ?personLabel ?falseDeath where {
  ?person wdt:P27 wd:Q29 .
  ?person p:P570 [
    ps:P570 ?falseDeath ;
    wikibase:rank wikibase:DeprecatedRank
  ] ;
SERVICE wikibase:label {
  bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".
}
}
```

Información errónea y cualificadores

- El motivo para que una tripleta sea deprecated se indica con la propiedad pq:2241

Personas que tienen una fecha de fallecimiento incorrecta debido a que están vivas

```
select ?person ?personLabel ?falseDeath where {
  ?person p:P570 [
    ps:P570 ?falseDeath ;
    pq:P2241 wd:Q21124171 ;
    wikibase:rank wikibase:DeprecatedRank
  ] ;
  SERVICE wikibase:label {
    bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en".
  }
}
```

Subconsultas

- Es posible hacer consultas internas para obtener resultados parciales

Población de países, junto a la media y la desviación típica de las poblaciones

```
SELECT ?paisLabel ?poblacion ?media (?poblacion - ?media as ?desv) {
  ?pais wdt:P31 wd:Q6256 .
  ?pais wdt:P1082 ?poblacion .
  { SELECT (AVG(?poblacion) as ?media) WHERE {
    ?pais wdt:P31 wd:Q6256 .
    ?pais wdt:P1082 ?poblacion .
  }
}
SERVICE wikibase:label {
  bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
} order by ?paisLabel
```

Consultas federadas

- Permiten combinar resultados de otros endpoints

Nombres de las obras de Lope de Vega, obtenidos del endpoint de la Biblioteca Virtual Cervantes

```
SELECT ?workLabel WHERE {  
  wd:Q165257 wdt:P2799 ?id  
  BIND(uri(concat("http://data.cervantesvirtual.com/person/", ?id)) as ?bvmcID)  
  SERVICE <http://data.cervantesvirtual.com/openrdf-sesame/repositories/data> {  
    ?bvmcID <http://rdaregistry.info/Elements/a/otherPFCManifestationOf> ?work .  
    ?work rdfs:label ?workLabel  
  }  
}
```

Lista de endpoints disponibles:

https://www.mediawiki.org/wiki/Special:MyLanguage/Wikidata_Query_Service/User_Manual/SPARQL_Federation_endpoints

Más información

- Tutorial SPARQL wikidata
 - https://www.wikidata.org/wiki/Wikidata:SPARQL_tutorial
- Para profundizar más: Libro SPARQL Wikidata
 - <https://en.wikibooks.org/wiki/SPARQL>
- Endpoint SPARQL with YASGUI
 - Example: <http://yasgui.org/short/WIfADkRSh>

Redirecciones y URIs

- La URI "de verdad" es re-dirigida a la URI HTML

Ejemplo:

<http://www.wikidata.org/entity/Q14317>

Se re-dirige a:

<https://www.wikidata.org/wiki/Q14317>

- Las propiedades también:

<https://www.wikidata.org/prop/direct/P1376>

Se re-dirige a: <https://www.wikidata.org/wiki/Property:P1376>

Permite separar concepto de su representación

En consultas es importante conocer la URI "de verdad"

Añadir datos a Wikidata

- Edición manual para ejemplos básicos/representativos
- Realizar pruebas con:
 - <http://test.wikidata.org>
 - <https://www.wikidata.org/wiki/Wikidata:Sandbox>
- Herramienta QuickStatements
 - <https://www.wikidata.org/wiki/Help:QuickStatements>
 - Añadir múltiples datos de una vez
- Wikidata Bots
 - <https://www.wikidata.org/wiki/Wikidata:Bots>

Wikibase: Crea tu propia Wikidata

- Wikibase = software para crear Wikidata
 - <https://wikiba.se/>
- Ofrece servicios de wikidata pero sin datos
 - MediaWiki
 - Editar, visualizar, consultar datos (SPARQL),
 - ...
- Puede utilizarse para gestionar datos propios
 - Permite edición controlada
 - Datos semánticos
 - Caso de uso habitual: Datos que no son CC0

Wikibase: ¿Cómo instalarlo?

- Instalación manual

- <https://www.mediawiki.org/wiki/Wikibase/Installation>

- Proceso largo

- Imágenes docker

- <https://github.com/wmde/wikibase-docker>

- Instalación muy rápida (recomendado)

- Utilizando wbstack

- <https://www.wbstack.com/>

- Mediante invitación de momento

Imágenes docker de wikibase

- <https://github.com/wmde/wikibase-docker>
- Instrucciones:
 - ▣ Instalar docker y descargar fichero docker-composer.yml
 - ▣ Ejecutar: `docker-compose up -d`
- Estará disponible en:
 - <http://localhost:8181> (Wikibase)
 - <http://localhost:8282> (Query service)
- Cerrar: `docker-compose stop`

Wikibase y endpoint SPARQL

- El endpoint SPARQL está en:
 - ▣ <URL>/proxy/wdqs/bigdata/namespace/wdq/sparql
 - ▣ <http://localhost:8282/proxy/wdqs/bigdata/namespace/wdq/sparql>
- Ejemplo de consulta SPARQL
 - ▣ <http://yasgui.org/short/qcuZI7OwL>
- URIs
 - ▣ Entidades: <http://wikibase.svc/entity/Q1>
 - ▣ Propiedades directas: <http://wikibase.svc/prop/direct/P1>

Esquemas de entidades

- En Mayo de 2019 se añade soporte a ShEx
- https://www.wikidata.org/wiki/Wikidata:WikiProject_Schemas
- Tutorial: https://www.wikidata.org/wiki/Wikidata:WikiProject_Schemas/Tutorial
- Formas de validación:
 - ▣ A través de la pagina de cada entidad
 - ▣ Herramienta RDFShape: <https://tinyurl.com/rq9hclu>
 - ▣ Herramienta WikiShape: <http://wikishape.weso.es/>

Esquemas de entidades en Wikibase

- La instancia wikibase también permite gestionar esquemas de entidades
- Ejemplo de validación local: <https://tinyurl.com/vbunkz9>

Datos lexicográficos

- Permite representar:
 - ▣ Lexemas
 - ▣ Formas (plural, femenino, etc.)
 - ▣ Significados (senses)
- Más información
 - ▣ https://www.wikidata.org/wiki/Wikidata:Lexicographical_data/Documentation
- Ejemplo (libro)
 - ▣ Libro: <https://www.wikidata.org/wiki/Lexeme:L317>

Más información

- Wikibase:

- <http://learningwikibase.com/>